<commit_before># ===========================================================================
# http://
# ===========================================================================
#
# SYNOPSIS
#
#   AX_UUID
#   AX_UUID_GENERATE_TIME_SAFE
#
# DESCRIPTION
#
#   Check for uuid, and uuid_generate_time_safe support.
#
# LICENSE
#
#   Copyright (c) 2012 <NAME> <<EMAIL>>
#
#   Copying and distribution of this file, with or without modification, are
#   permitted in any medium without royalty provided the copyright notice
#   and this notice are preserved. This file is offered as-is, without any
#   warranty.

#serial 3

AC_DEFUN([AX_UUID], [
         AC_CHECK_HEADER([uuid/uuid.h], [
                         AC_CACHE_CHECK([check to see if -luuid is needed], [ax_cv_libuuid_is_required], [
                                        AC_LANG_PUSH([C])
                                        AC_COMPILE_IFELSE([
                                                          AC_LANG_PROGRAM([#include <uuid/uuid.h>], [
                                                                          uuid_t out;
                                                                          uuid_generate(out);
                                                                          ])],
                                                          [ax_cv_libuuid_is_required=no],
                                                          [ax_cv_libuuid_is_required=maybe])
                                        AC_LANG_POP
                                        ])

                         AS_IF([test "$ax_cv_libuuid_is_required" = maybe], [
                               AC_CACHE_CHECK([check to see if -luuid is not needed], [ax_cv_libuuid_works], [
                                              save_LDFLAGS="$LDFLAGS"
                                              LDFLAGS="-luuid"
                                              AC_LANG_PUSH([C])
                                              AC_LINK_IFELSE([
                                                             AC_LANG_PROGRAM([#include <uuid/uuid.h>], [
                                                                             uuid_t out;
                                                                             uuid_generate(out);
                                                                             ])],
                                                             [ax_cv_libuuid_works=yes])
                                              AC_LANG_POP
                                              LDFLAGS="$save_LDFLAGS"
                                              ])
                               AS_IF([test "$ax_cv_libuuid_works" = yes], [LIBUUID_LDFLAGS="-luuid"])
                               ])

                         AS_IF([test "$ax_cv_libuuid_is_required" = no], [UUID_UUID_H=yes])
                         AS_IF([test "$ax_cv_libuuid_works" = yes], [UUID_UUID_H=yes])
                         ])

AS_IF([test "$UUID_UUID_H" = yes], [
      AC_DEFINE([HAVE_UUID_UUID_H], [ 1 ], [Have uuid/uuid.h])
      ],[
      AC_DEFINE([HAVE_UUID_UUID_H], [ 0 ], [Have uuid/uuid.h])
      ])
])

AC_DEFUN([AX_UUID_GENERATE_TIME_SAFE], [
         AC_REQUIRE([AX_UUID])
         AC_CACHE_CHECK([for uuid_generate_time_safe], [ax_cv_uuid_generate_time_safe], [
                        save_LDFLAGS="$LDFLAGS"
                        LDFLAGS="-luuid"
                        AC_LANG_PUSH([C])
                        AC_LINK_IFELSE([
                                       AC_LANG_PROGRAM([#include <uuid/uuid.h>], [
                                                       uuid_t out;
                                                       uuid_generate_time_safe(out);
                                                       ])],
                                       [ax_cv_uuid_generate_time_safe=yes],
                                       [ax_cv_uuid_generate_time_safe=no])
                        AC_LANG_POP
                        LDFLAGS="$save_LDFLAGS"
                        ])

  AS_IF([test "$ax_cv_uuid_generate_time_safe" = yes],[
        AC_SUBST([LIBUUID_LDFLAGS],[-luuid])
        AC_DEFINE(HAVE_UUID_GENERATE_TIME_SAFE,[1],[Define if uuid_generate_time_safe is present in uuid/uuid.h.])],[
        AC_DEFINE(HAVE_UUID_GENERATE_TIME_SAFE,[0],[Define if uuid_generate_time_safe is present in uuid/uuid.h.])
        ])
])<commit_msg>Test with run, not link.<commit_after># ===========================================================================
# http://
# ===========================================================================
#
# SYNOPSIS
#
#   AX_UUID
#   AX_UUID_GENERATE_TIME_SAFE
#
# DESCRIPTION
#
#   Check for uuid, and uuid_generate_time_safe support.
#
# LICENSE
#
#   Copyright (c) 2012 <NAME> <<EMAIL>.org>
#
#   Copying and distribution of this file, with or without modification, are
#   permitted in any medium without royalty provided the copyright notice
#   and this notice are preserved. This file is offered as-is, without any
#   warranty.

#serial 3

AC_DEFUN([AX_UUID], [
    AC_CHECK_HEADER([uuid/uuid.h], [
      AC_CACHE_CHECK([check to see if -luuid is needed], [ax_cv_libuuid_is_required], [
        AC_LANG_PUSH([C])
        AC_RUN_IFELSE([
          AC_LANG_PROGRAM([#include <uuid/uuid.h>], [
            uuid_t out;
            uuid_generate(out);
            return 0;
            ])],
          [ax_cv_libuuid_is_required=no],
          [ax_cv_libuuid_is_required=maybe])
        AC_LANG_POP
        ])

      AS_IF([test "$ax_cv_libuuid_is_required" = maybe], [
        AC_CACHE_CHECK([check to see if -luuid is not needed], [ax_cv_libuuid_works], [
          save_LDFLAGS="$LDFLAGS"
          LDFLAGS="-luuid"
          AC_LANG_PUSH([C])
          AC_RUN_IFELSE([
            AC_LANG_PROGRAM([#include <uuid/uuid.h>], [
              uuid_t out;
              uuid_generate(out);
              return 0;
              ])],
            [ax_cv_libuuid_works=yes])
          AC_LANG_POP
          LDFLAGS="$save_LDFLAGS"
          ])
        AS_IF([test "$ax_cv_libuuid_works" = yes], [
          AC_SUBST([LIBUUID_LDFLAGS],[-luuid])])
        ])

      AS_IF([test "$ax_cv_libuuid_is_required" = no], [UUID_UUID_H=yes])
      AS_IF([test "$ax_cv_libuuid_works" = yes], [UUID_UUID_H=yes])
      ])

  AS_IF([test "$UUID_UUID_H" = yes], [
      AC_DEFINE([HAVE_UUID_UUID_H], [ 1 ], [Have uuid/uuid.h])
      ],[
      AC_DEFINE([HAVE_UUID_UUID_H], [ 0 ], [Have uuid/uuid.h])
      ])
  ])

  AC_DEFUN([AX_UUID_GENERATE_TIME_SAFE], [
      AC_REQUIRE([AX_UUID])
      AC_CACHE_CHECK([for uuid_generate_time_safe], [ax_cv_uuid_generate_time_safe], [
        save_LDFLAGS="$LDFLAGS"
        LDFLAGS="-luuid"
        AC_LANG_PUSH([C])
        AC_LINK_IFELSE([
          AC_LANG_PROGRAM([#include <uuid/uuid.h>], [
            uuid_t out;
            uuid_generate_time_safe(out);
            ])],
          [ax_cv_uuid_generate_time_safe=yes],
          [ax_cv_uuid_generate_time_safe=no])
        AC_LANG_POP
        LDFLAGS="$save_LDFLAGS"
        ])

      AS_IF([test "$ax_cv_uuid_generate_time_safe" = yes],[
        AC_SUBST([LIBUUID_LDFLAGS],[-luuid])
        AC_DEFINE(HAVE_UUID_GENERATE_TIME_SAFE,[1],[Define if uuid_generate_time_safe is present in uuid/uuid.h.])],[
        AC_DEFINE(HAVE_UUID_GENERATE_TIME_SAFE,[0],[Define if uuid_generate_time_safe is present in uuid/uuid.h.])
        ])
      ])
{'fraction_non_alphanumeric': 0.1359784801613988, 'fraction_numerical': 0.0026899798251513113, 'mean_word_length': 1.4284781188765512, 'pattern_counts': {'":': 0, '<': 15, '<?xml version=': 0, '>': 15, 'https://': 0, 'lorem ipsum': 0, 'www.': 0, 'xml': 0}, 'pii_count': 0, 'substrings_counts': 0, 'word_list_counts': {'cursed_substrings.json': 55, 'profanity_word_list.json': 0, 'sexual_word_list.json': 0, 'zh_pornsignals.json': 0}}
{'dir': 'git-commits-cleaned', 'id': '6379975', 'n_tokens_mistral': 2096, 'n_tokens_neox': 2045, 'n_words': 322}